

# FUNCTIONAL PEARLS

## *The Bird Tree*

RALF HINZE

*Computing Laboratory, University of Oxford, Wolfson Building, Parks Road, Oxford, OX1 3QD, England*  
(e-mail: ralf.hinze@comlab.ox.ac.uk)

### 1 Introduction

Sadly, Richard Bird is stepping down as the editor of the ‘Functional Pearls’ column. As a farewell present, I would like to dedicate a tree to him. A woody plant is appropriate for at least two reasons: Richard has been preoccupied with trees in many of his pearls, and where else would you find a bird’s nest? Actually, there is a lot of room for nests, as the tree is infinite. Figure 1 displays the first five levels. The Bird tree, whose nodes are labelled with rational numbers, enjoys several remarkable properties.

Firstly, it is a *fractal* object, in the sense that parts of it are similar to the whole. The Bird tree can be transformed into its left subtree by first incrementing and then reciprocalsing the elements. To obtain the right subtree, we have to interchange the order of the two steps: the elements are first reciprocalsed and then incremented. This description can be nicely captured by a *co-recursive* definition, given in the purely functional programming language Haskell (Peyton Jones 2003):

$$\begin{aligned} \text{bird} &:: \text{Tree Rational} \\ \text{bird} &= \text{Node } 1 \ (1 / (\text{bird} + 1)) \ ((1 / \text{bird}) + 1). \end{aligned}$$

The definitions that make this work are introduced in the next section. For the moment, it suffices to know that the arithmetic operations are lifted pointwise to trees. For instance,  $\text{bird} + 1$  is the same as  $\text{map } (+1) \ \text{bird}$ .

Returning to the tree properties, the picture suggests that mirroring the tree yields its reciprocal,  $\text{mirror } \text{bird} = 1 / \text{bird}$ , and this is indeed the case. Furthermore, consider the sequence of rationals along the left (or the right) spine of the Bird tree. We discover some old friends: each fraction consists of two consecutive *Fibonacci* numbers. In other words, we approximate the *golden ratio*  $\phi = (1 + \sqrt{5})/2$  as we go down the right spine. The tree also contains the natural numbers. For those, we have to descend in a zigzag fashion: right, left, right, left and so forth. On the other hand, if we list the numerators (or denominators) level-wise, we obtain a somewhat obscure sequence, which is not even listed in Sloane’s ‘On-Line Encyclopedia of Integer Sequences’ (2009).<sup>1</sup>

<sup>1</sup> I have submitted the sequences, numerators and denominators of *bird* and its bit-reversal permutation tree (see Section 3) to the ‘On-Line Encyclopedia of Integer Sequences’ (preliminary A-numbers: A162909–A162912).




Fig. 1. The Bird tree.

The most intriguing property of the Bird tree, however, is the following: Like the Stern–Brocot tree (Graham *et al.* 1994) or the Calkin–Wilf tree (Calkin & Wilf 2000), it enumerates all the positive rationals. In other words, the tree contains every positive rational exactly once.

The purpose of this pearl is twofold. First, we shall, of course, justify the claims made above. Sections 2 and 3 work towards this goal, reviewing the main proof technique and relating recursive and iterative tree definitions. Section 4 then shows that the Bird tree and the Stern–Brocot tree are level-wise permutations of each other. Second, we aim to derive a loopless algorithm for linearising the Bird tree. Section 6 develops a general algorithm, with Section 5 preparing the ground.

### 2 Infinite trees, idioms and unique fixed points

In a lazy functional language such as Haskell, infinite trees are easy to define:

```
data Tree α = Node {root :: α, left :: Tree α, right :: Tree α}.
```

The type *Tree* α is a so-called co-inductive datatype. Its definition is similar to the standard textbook definition of binary trees, except that there is no base constructor, so we cannot build a finite tree. Since there is no base case, *mirror* is a one-liner:

```
mirror                :: Tree α → Tree α
mirror (Node a l r) = Node a (mirror r) (mirror l).
```

The function *mirror* like many more to come relies critically on lazy evaluation.

The definition of *bird* uses + and / lifted to trees. We obtain these liftings almost for free, as *Tree* is a so-called applicative functor or idiom (McBride & Paterson 2008):

```
infixl 9 ◇
class Idiom φ where
  pure :: α → φ α
  (◇)  :: φ (α → β) → (φ α → φ β)

instance Idiom Tree where
  pure a = t where t = Node a t t
  t ◇ u  = Node ((root t) (root u)) (left t ◇ left u) (right t ◇ right u).
```

The call  $\text{pure } a$  constructs an infinite tree of  $a$ s; the idiomatic application  $\diamond$  takes a tree of functions and a tree of arguments to a tree of results.

Every instance of *Idiom* must satisfy

$$\begin{aligned} \text{pure } id \diamond u &= u && \text{(identity)} \\ \text{pure } (\circ) \diamond u \diamond v \diamond w &= u \diamond (v \diamond w) && \text{(composition)} \\ \text{pure } f \diamond \text{pure } x &= \text{pure } (f \ x) && \text{(homomorphism)} \\ u \diamond \text{pure } x &= \text{pure } (\lambda f \rightarrow f \ x) \diamond u, && \text{(interchange)} \end{aligned}$$

which allow us to rewrite every idiom expression into the form  $\text{pure } f \diamond a_1 \diamond \dots \diamond a_n$ . So idioms capture the idea of applying a pure function to ‘impure’ arguments.

We single out two special cases that we will need time and again:  $\text{map } f \ t = \text{pure } f \diamond t$  and  $\text{zip } g \ t \ u = \text{pure } g \diamond t \diamond u$ . The function  $\text{zip}$  lifts a binary operator to an idiomatic structure; for instance,  $(*) = \text{zip } (,)$  turns a pair of trees into a tree of pairs. In general,  $\text{pure } f \diamond a_1 \diamond \dots \diamond a_n$  lifts an  $n$ -ary function pointwise to an idiomatic structure. Using this ‘idiom’ we can define a generic instance of *Num*:<sup>2</sup>

**instance** (*Idiom*  $\phi$ , *Num*  $\alpha$ )  $\Rightarrow$  *Num* ( $\phi \ \alpha$ ) **where**

$$\begin{aligned} (+) &= \text{zip } (+) \\ (-) &= \text{zip } (-) \\ (*) &= \text{zip } (*) \\ \text{negate} &= \text{map } \text{negate} \\ \text{fromInteger} &= \text{pure } \circ \text{fromInteger}. \end{aligned}$$

In this pearl, we consider two idioms, infinite trees and streams. In both cases, the familiar arithmetic laws also hold for the lifted operators.

Every structure comes equipped with structure-preserving maps; so do idioms: a map  $h :: \phi \ \alpha \rightarrow \psi \ \alpha$  is an *idiom homomorphism* iff

$$h (\text{pure } a) = \text{pure } a \tag{1}$$

$$h (x \diamond y) = h \ x \diamond h \ y. \tag{2}$$

The map *mirror* is an example of an idiom homomorphism; it is even an *idiom isomorphism*, since  $\text{mirror} \circ \text{mirror} = id$ . This fact greatly simplifies reasoning, as we can, for instance, effortlessly rewrite  $\text{mirror} ((1 / \text{bird}) + 1) = \text{mirror} (\text{pure } (+) \diamond (\text{pure } (/) \diamond \text{pure } 1 \diamond \text{bird}) \diamond \text{pure } 1)$  to  $\text{pure } (+) \diamond (\text{pure } (/) \diamond \text{pure } 1 \diamond \text{mirror } \text{bird}) \diamond \text{pure } 1 = (1 / \text{mirror } \text{bird}) + 1$ .

This is all very well, but how do we prove the idiom and the homomorphism laws in the first place? It turns out that the type of infinite trees enjoys an attractive and easy-to-use proof principle. Consider the recursion equation  $x = \text{Node } a \ l \ r$ , where  $l$  and  $r$  possibly contain the variable  $x$  but *not* the expressions *root*  $x$ , *left*  $x$  or *right*  $x$ . Equations of this syntactic form possess a *unique solution*. (Rutten (2003) shows an analogous statement for streams; the proof, however, can be readily adapted to infinite trees.) Uniqueness can be exploited to prove that two infinite

<sup>2</sup> Unfortunately, this doesn’t quite work with the Standard Haskell libraries, as *Num* has two superclasses, *Eq* and *Show*, which can’t sensibly be defined generically.

trees are equal: if they satisfy the same recursion equation, then they are. The proof of  $1 / bird = mirror\ bird$  illustrates the idea:

$$\begin{aligned}
 & 1 / bird \\
 = & \quad \{ \text{definition of } bird \} \\
 & 1 / Node\ 1\ (1 / (bird + 1))\ ((1 / bird) + 1) \\
 = & \quad \{ \text{arithmetic} \} \\
 & Node\ 1\ ((1 / (1 / bird)) + 1)\ (1 / ((1 / bird) + 1)) \\
 \subset & \quad \{ x = Node\ 1\ ((1 / x) + 1)\ (1 / (x + 1)) \text{ has a unique solution} \} \\
 & Node\ 1\ ((1 / mirror\ bird) + 1)\ (1 / (mirror\ bird + 1)) \\
 = & \quad \{ mirror\ \text{is an idiom homomorphism} \} \\
 & Node\ 1\ (mirror\ ((1 / bird) + 1))\ (mirror\ (1 / (bird + 1))) \\
 = & \quad \{ \text{definition of } mirror\ \text{and } bird \} \\
 & mirror\ bird.
 \end{aligned}$$

The link  $\subset$  indicates that the proof rests on the *unique fixed-point principle*; the recursion equation is given within the curly braces. The upper part shows that  $1 / bird$  satisfies the equation  $x = Node\ 1\ ((1 / x) + 1)\ (1 / (x + 1))$ ; the lower part establishes that  $mirror\ bird$  satisfies the same equation. The symbol  $\subset$  links the two parts, effectively proving the equality of both expressions. As regards contents, the proof relies on the facts that 1 is a fixed point of the reciprocal function and that reciprocal is an *involution*.

**Exercise 1** Using the unique fixed-point principle, show that ‘Tree’ satisfies the idiom laws and that ‘mirror’ is an idiom homomorphism.

### 3 Recursion and iteration

The combinator *recurse* captures *recursive* or *top-down* tree constructions; the functions  $f$  and  $g$  are repeatedly mapped over the whole tree:

$$\begin{aligned}
 & recurse \quad :: (\alpha \rightarrow \alpha) \rightarrow (\alpha \rightarrow \alpha) \rightarrow (\alpha \rightarrow Tree\ \alpha) \\
 & recurse\ f\ g\ a = t \ \mathbf{where}\ t = Node\ a\ (map\ f\ t)\ (map\ g\ t).
 \end{aligned}$$

Thus, an alternative definition of *bird* is  $recurse\ (recip \circ succ)\ (succ \circ recip)\ 1$ , where *recip* is the reciprocal function and *succ* is the successor function.

We can also construct a tree in an *iterative* or *bottom-up* fashion; the functions  $f$  and  $g$  are repeatedly applied to the given initial seed  $a$ :

$$\begin{aligned}
 & iterate \quad :: (\alpha \rightarrow \alpha) \rightarrow (\alpha \rightarrow \alpha) \rightarrow (\alpha \rightarrow Tree\ \alpha) \\
 & iterate\ f\ g\ a = loop\ a \ \mathbf{where}\ loop\ x = Node\ x\ (loop\ (f\ x))\ (loop\ (g\ x)).
 \end{aligned}$$

The type  $\alpha$  can be seen as a type of states and the infinite tree as an enumeration of the state space. One could argue that *iterate* is more natural than *recurse*. This intuition is backed up by the fact that  $map\ h \circ iterate\ f\ g$  is the unfold of the *Tree* co-datatype.




Fig. 2. A tree that contains all bit strings and its bit-reversal permutation tree.

The goal of this section is to turn a recursive definition, such as the one for *bird*, into an iterative one, which can be executed manually to grow a tree. Before we tackle this problem, we note that both *recurse* and *iterate* satisfy a *fusion law*:

$$\begin{array}{rcc}
 \text{map } h \circ \text{recurse } f_1 \ g_1 & = & \text{recurse } f_2 \ g_2 \circ h \\
 & \uparrow & \\
 h \circ f_1 = f_2 \circ h & \wedge & h \circ g_1 = g_2 \circ h \\
 & \downarrow & \\
 \text{map } h \circ \text{iterate } f_1 \ g_1 & = & \text{iterate } f_2 \ g_2 \circ h.
 \end{array}$$

**Exercise 2** Prove the fusion laws, and then use fusion to give an alternative proof that  $1 / \text{bird} = \text{mirror bird}$ .

How are *recurse*  $f \ g \ a$  and *iterate*  $f \ g \ a$  related? Consider Figure 2, which displays the trees *recurse*  $([0]++) \ ([1]++) \ []$  and *iterate*  $([0]++) \ ([1]++) \ []$ . Since  $f$  and  $g$  are applied in different orders – inside out and outside in – each level of *recurse*  $f \ g \ a$  is the *bit-reversal permutation* of the corresponding level of *iterate*  $f \ g \ a$ . For brevity’s sake, one tree is called the *bit-reversal permutation tree* of the other. Can we transform an instance of *recurse* into an instance of *iterate*? Yes, if the two functions are pre- or post-multiplications of elements of some given *monoid*. Let us introduce a suitable type class:

```

infixr 5 ·
class Monoid α where
  ε :: α
  (·) :: α → α → α.
    
```

The *recursion–iteration lemma* then states

$$\text{recurse } (a \cdot) (b \cdot) \epsilon = \text{iterate } (\cdot a) (\cdot b) \epsilon, \tag{3}$$

where  $a$  and  $b$  are elements of some monoid  $(M, \cdot, \epsilon)$ . To establish the lemma, we show that  $\text{iterate } (\cdot a) (\cdot b) \epsilon$  satisfies the defining equation of  $\text{recurse } (a \cdot) (b \cdot) \epsilon$ , that is  $t = \text{Node } \epsilon (\text{map } (a \cdot) t) (\text{map } (b \cdot) t)$ :

$$\begin{aligned} & \text{iterate } (\cdot a) (\cdot b) \epsilon \\ = & \quad \{ \text{definition of } \text{iterate} \} \\ & \text{Node } \epsilon (\text{iterate } (\cdot a) (\cdot b) (\epsilon \cdot a)) (\text{iterate } (\cdot a) (\cdot b) (\epsilon \cdot b)) \\ = & \quad \{ \epsilon \cdot x = x = x \cdot \epsilon \} \\ & \text{Node } \epsilon (\text{iterate } (\cdot a) (\cdot b) (a \cdot \epsilon)) (\text{iterate } (\cdot a) (\cdot b) (b \cdot \epsilon)) \\ = & \quad \{ \text{fusion: } (x \cdot) \circ (\cdot y) = (\cdot y) \circ (x \cdot) \} \\ & \text{Node } \epsilon (\text{map } (a \cdot) (\text{iterate } (\cdot a) (\cdot b) \epsilon)) (\text{map } (b \cdot) (\text{iterate } (\cdot a) (\cdot b) \epsilon)). \end{aligned}$$

At first sight, it seems that the applicability of the lemma is somewhat hampered by the requirement on the form of the two arguments. However, since *endomorphisms*, functions of type  $\tau \rightarrow \tau$  for some  $\tau$ , form a monoid, we can easily rewrite an arbitrary instance of *recurse* into the required form ( $\diamond$  is function application below, the ‘apply’ of the identity idiom):

$$\begin{aligned} & \text{recurse } (\text{recip} \circ \text{succ}) (\text{succ} \circ \text{recip}) 1 \\ = & \quad \{ \text{fusion: } \text{id} \diamond x = x \text{ and } (\diamond x) \circ (f \circ) = f \circ (\diamond x) \} \\ & \text{recurse } (\text{recip} \circ \text{succ} \circ) (\text{succ} \circ \text{recip} \circ) \text{id} \diamond 1 \\ = & \quad \{ \text{recursion-iteration lemma} \} \\ & \text{iterate } (\circ \text{recip} \circ \text{succ}) (\circ \text{succ} \circ \text{recip}) \text{id} \diamond 1. \end{aligned}$$

Hooray, we have succeeded in transforming *bird* into an iterative form! Well, not quite; one could argue that using functions as the ‘internal state’ is cheating. Fortunately, we can provide a concrete representation of these functions by viewing a rational as a pair of numbers. To this end, we introduce a type of vectors:

$$\mathbf{data} \text{ Vector} = \begin{pmatrix} \text{Integer} \\ \text{Integer} \end{pmatrix}; \quad \mathbf{i} = \begin{pmatrix} 1 \\ 1 \end{pmatrix}.$$

The function *rat* maps the concrete to the abstract representation:

$$\begin{aligned} \text{rat} & \quad \text{:: } \text{Vector} \rightarrow \text{Rational} \\ \text{rat} \begin{pmatrix} a \\ b \end{pmatrix} & = a \div b, \end{aligned}$$

where  $\div$  constructs a rational from two integers.

Both *recip* and *succ* can be easily expressed as vector transformations. In fact, since they correspond to linear transformations, we can phrase them as matrix multiplications:

$$\mathbf{data} \text{ Matrix} = \begin{pmatrix} \text{Integer} & \text{Integer} \\ \text{Integer} & \text{Integer} \end{pmatrix}.$$






Fig. 3. The Stern-Brocot tree

**Exercise 3** Formalise the claim above. You may want to peek at Section 5 first, which introduces infinite lists. (Hint: define a function  $spine :: Tree\ \alpha \rightarrow Stream\ \alpha$  that maps a tree on to its left or right spine.)

### 4 The Stern–Brocot tree

There are many ways to enumerate the positive rationals. Probably the oldest method was found around 1850 by the German mathematicians Eisenstein and Stern. It is deceptively simple: start with the two ‘boundary rationals’  $0/1$  and  $1/0$ , which are not included in the enumeration, and then repeatedly insert the median  $a+b/c+d$  between two adjacent rationals  $a/c$  and  $b/d$ .

Since the number of inserted rationals doubles with every step, the process can be pictured by an infinite binary tree, the so-called Stern–Brocot tree<sup>3</sup> (see Figure 3). Quite remarkably, each level shown is a permutation of the corresponding level of the Bird tree. The purpose of this section is to verify this observation, which implies that the Bird tree also contains every positive rational once.

Before we work out the relationship, let us turn the informal description of the Stern–Brocot tree into a program. This is most easily accomplished if we first construct a tree of intervals, represented by  $2 \times 2$  matrices, and then map the intervals to their medians. The start interval is now  $\begin{pmatrix} 0 & 1 \\ 1 & 0 \end{pmatrix}$ ; the interval is updated as follows:

$$\begin{pmatrix} a & b \\ c & d \end{pmatrix} * \begin{matrix} \blacksquare \\ \blacksquare \end{matrix} = \begin{pmatrix} a & a+b \\ c & c+d \end{pmatrix} \leftarrow \begin{pmatrix} a & b \\ c & d \end{pmatrix} \mapsto \begin{pmatrix} a+b & b \\ c+d & d \end{pmatrix} = \begin{pmatrix} a & b \\ c & d \end{pmatrix} * \begin{matrix} \blacksquare \\ \blacksquare \end{matrix}.$$

The left bound of the left descendent is the original left bound; the right bound is the median of the two original bounds. Likewise for the right descendent.

So the verbal description corresponds to an iterative construction, in which the state is an interval. Using a derivation inverse to the one in the preceding section, we can turn the verbal description into a compact co-recursive definition:

$$\begin{aligned} & \text{map median (iterate (*\blacksquare) (*\blacksquare) F)} \\ = & \quad \{ \text{fusion: } \mathbf{I} * \mathbf{F} = \mathbf{F}, \mathbf{F} * \blacksquare = \blacksquare * \mathbf{F} \text{ and } \mathbf{F} * \blacksquare = \blacksquare * \mathbf{F} \} \\ & \text{map median (map (*F) (iterate (*\blacksquare) (*\blacksquare) \mathbf{I}))} \end{aligned}$$

<sup>3</sup> The French clockmaker Brocot discovered the method around 1860, independent of Eisenstein and Stern.



$$\begin{aligned}
 &= \{ \text{functor and } \textit{mediant} \circ (*\mathbf{F}) = \textit{mediant} \} \\
 &\quad \textit{map mediant} (\textit{iterate} (*\mathbf{I}) (*\mathbf{I}) \mathbf{I}) \\
 &= \{ \text{recursion-iteration lemma} \} \\
 &\quad \textit{map mediant} (\textit{recurse} (\mathbf{I}*) (\mathbf{I}*) \mathbf{I}) \\
 &= \{ \text{fusion: } \mathbf{I} * \mathbf{v} = \mathbf{v} \text{ and } (*\mathbf{v}) \circ (\mathbf{X}*) = (\mathbf{X}*) \circ (*\mathbf{v}) \} \\
 &\quad \textit{map rat} (\textit{recurse} (\mathbf{I}*) (\mathbf{I}*) \mathbf{i}) \\
 &= \{ \mathbf{F} * \mathbf{I} * \mathbf{F} = \mathbf{I} \} \\
 &\quad \textit{map rat} (\textit{recurse} (\mathbf{F} * \mathbf{I} * \mathbf{F}) (\mathbf{I}*) \mathbf{i}) \\
 &= \{ \text{fusion: } \textit{rat} \mathbf{i} = 1, (4) \text{ and } (5) \} \\
 &\quad \textit{recurse} (\textit{recip} \circ \textit{succ} \circ \textit{recip}) \textit{succ} 1.
 \end{aligned}$$

If we unfold the definitions, we obtain the following co-recursive program:

$$\begin{aligned}
 \textit{stern-brocot} &:: \textit{Tree Rational} \\
 \textit{stern-brocot} &= \textit{Node} 1 (1 / (1 / \textit{stern-brocot} + 1)) (\textit{stern-brocot} + 1).
 \end{aligned}$$

The definition is tantalisingly close to the definition of *bird*. As an aside, the derivation above also provides a formula for the bit-reversal permutation tree of *stern-brocot*, the so-called Calkin–Wilf or Eisenstein–Stern tree. We simply replace *recurse* by *iterate*, obtaining *iterate (recip ∘ succ ∘ recip) succ 1*.

We have already observed that *bird* and *stern-brocot* are level-wise permutations of each other. Looking a bit closer, we notice that the natural numbers are located on the right spine of the Stern–Brocot tree, whereas the Fibonacci fractions that approach the golden ratio  $1/1, 2/1, 3/2, 5/3, 5/8$ , etc. appear on a zigzag path. Recalling that it was the other way round in the Bird tree, we conjecture

$$\textit{odd-mirror bird} = \textit{stern-brocot} \tag{6}$$

$$\textit{odd-mirror stern-brocot} = \textit{bird}, \tag{7}$$

where *odd-mirror* swaps the immediate subtrees of a node but only on odd levels:

$$\begin{aligned}
 \textit{even-mirror, odd-mirror} &:: \textit{Tree } \alpha \rightarrow \textit{Tree } \alpha \\
 \textit{odd-mirror} (\textit{Node} a l r) &= \textit{Node} a (\textit{even-mirror} l) (\textit{even-mirror} r) \\
 \textit{even-mirror} (\textit{Node} a l r) &= \textit{Node} a (\textit{odd-mirror} r) (\textit{odd-mirror} l).
 \end{aligned}$$

Since *odd-mirror* and *even-mirror* are involutions, it suffices to prove one of the equations above; we pick the first one (6). Let us introduce some shortcuts so that the expressions still fit on one line: We abbreviate *map recip* by *r*, *map (recip ∘ succ)* by *rs* and so forth. Furthermore, *e* is shorthand for *even-mirror bird* and likewise *o* for *odd-mirror bird*. Finally, we abbreviate *stern-brocot* by *sb*:

$$\begin{aligned}
 &o \\
 &= \{ \text{definition of } \textit{bird} \text{ and } \textit{odd-mirror}, \text{ naturality of } \textit{even-mirror} \} \\
 &\quad \textit{Node} 1 (\textit{rs} e) (\textit{sr} e)
 \end{aligned}$$

$$\begin{aligned}
&= \{ \text{definition of } \textit{bird} \text{ and } \textit{even-mirror}, \text{ naturality of } \textit{odd-mirror} \} \\
&\quad \textit{Node } 1 \text{ (} rs \text{ (} \textit{Node } 1 \text{ (} sr \text{ } o \text{)) (} rs \text{ } o \text{)) (} sr \text{ (} \textit{Node } 1 \text{ (} sr \text{ } o \text{)) (} rs \text{ } o \text{))} \\
&= \{ \text{definition of } rs \text{ and } sr \} \\
&\quad \textit{Node } 1 \text{ (} \textit{Node } ^{1/2} \text{ (} rssr \text{ } o \text{) (} rsrs \text{ } o \text{)) (} \textit{Node } ^{2/1} \text{ (} srsr \text{ } o \text{) (} srss \text{ } o \text{))} \\
&\subset \{ x = \textit{Node } 1 \text{ (} \textit{Node } ^{1/2} \text{ (} rssr \text{ } x \text{) (} rsrs \text{ } x \text{)) (} \textit{Node } ^{2/1} \text{ (} srsr \text{ } x \text{) (} srss \text{ } x \text{))} \} \\
&\quad \textit{Node } 1 \text{ (} \textit{Node } ^{1/2} \text{ (} rssr \text{ } sb \text{) (} rsrs \text{ } sb \text{)) (} \textit{Node } ^{2/1} \text{ (} srsr \text{ } sb \text{) (} srss \text{ } sb \text{))} \\
&= \{ \textit{recip} \text{ is an involution: } rssr = rsrrsr \text{ and } srss = ss \} \\
&\quad \textit{Node } 1 \text{ (} \textit{Node } ^{1/2} \text{ (} rsrrsr \text{ } sb \text{) (} rsrs \text{ } sb \text{)) (} \textit{Node } ^{2/1} \text{ (} srsr \text{ } sb \text{) (} ss \text{ } sb \text{))} \\
&= \{ \text{definition of } rsr \text{ and } s \} \\
&\quad \textit{Node } 1 \text{ (} rsr \text{ (} \textit{Node } 1 \text{ (} rsr \text{ } sb \text{)) (} s \text{ } sb \text{)) (} s \text{ (} \textit{Node } 1 \text{ (} rsr \text{ } sb \text{)) (} s \text{ } sb \text{))} \\
&= \{ \text{definition of } sb \} \\
&\quad \textit{Node } 1 \text{ (} rsr \text{ } sb \text{) (} s \text{ } sb \text{)} \\
&= \{ \text{definition of } sb \} \\
&\quad sb.
\end{aligned}$$

## 5 Linearising the Stern–Brocot tree

Now, let us consider linearising the Bird tree. As a warm-up exercise, this section demonstrates how to linearise the Stern–Brocot tree. This has been done several times before (Gibbons *et al.* 2006; Backhouse & Ferreira 2008), but we believe that the co-data framework is particularly suited to this task, so it is worthwhile to repeat the exercise. Technically, we aim to derive a *loopless algorithm* (Bird 2006) for enumerating the elements of *stern-brocot*. An enumeration is called *loopless* if the next element is computed from the previous one in constant time and, for this pearl, in constant space.

Since we have to transform an infinite tree into an infinite list, let us introduce a tailor-made co-datatype for the latter (Hinze 2008):

```

data Stream  $\alpha$  = Cons { head ::  $\alpha$ , tail :: Stream  $\alpha$  }

infixr 5 <
(<)  ::  $\alpha \rightarrow$  Stream  $\alpha \rightarrow$  Stream  $\alpha$ 
a < s = Cons a s.

```

The type of streams is similar to Haskell’s predefined type of lists, except that there is no empty list constructor, so we cannot form a finite list.

Like infinite trees, streams are an idiom, which means that we can effortlessly lift functions to streams:

```

instance Idiom Stream where
  pure a = s where s = a < s
  s  $\diamond$  t = (head s) (head t) < (tail s)  $\diamond$  (tail t).

```

Like infinite trees, streams can be built recursively or iteratively. We overload *recurse*

and *iterate* to also denote the combinators on streams:

$$\begin{aligned} \text{recurse} &:: (\alpha \rightarrow \alpha) \rightarrow (\alpha \rightarrow \text{Stream } \alpha) \\ \text{recurse } f \ a &= s \ \mathbf{where} \ s = a < \text{map } f \ s \\ \text{iterate} &:: (\alpha \rightarrow \alpha) \rightarrow (\alpha \rightarrow \text{Stream } \alpha) \\ \text{iterate } f \ a &= a < \text{iterate } f \ (f \ a). \end{aligned}$$

Unlike the tree combinators, *recurse* and *iterate* construct exactly the same stream: we have  $\text{recurse } f \ a = \text{iterate } f \ a$ .

**Exercise 4** Show that  $\text{iterate } f \ a$  satisfies the recursion equation of  $\text{recurse } f \ a$ . (Hint: Formulate a fusion law first.)

To convert a tree to a stream, we define a helper function that chops the root off a tree:

$$\begin{aligned} \text{stream} &:: \text{Tree } \alpha \rightarrow \text{Stream } \alpha \\ \text{stream } t &= \text{root } t < \text{stream } (\text{chop } t) \\ \text{chop} &:: \text{Tree } \alpha \rightarrow \text{Tree } \alpha \\ \text{chop } t &= \text{Node } (\text{root } (\text{left } t)) (\text{right } t) (\text{chop } (\text{left } t)). \end{aligned}$$

In a sense, *root* is the counterpart of *head* and *chop* is the counterpart of *tail*. Infinite trees and streams are both very regular structures, so it probably comes as little surprise that *stream* is an idiom isomorphism.

**Exercise 5** Show that ‘*stream*’ is an idiom isomorphism and that ‘*chop*’ is an idiom homomorphism.

Let’s have a closer look at the workings of *stream*: it outputs the elements of its argument tree level by level. However, since *chop* repeatedly swaps the left and the right subtree, each level is output in *bit-reversal permutation* order. In other words, *stream stern-brocot* actually linearises the Calkin–Wilf tree. We return to this point later on. The enumeration  $\text{stream } t$  is not loopless: to produce the next element,  $\text{stream } t$  takes time linear in the depth of the element in the tree and space proportional to the width of the current level. So turning *stream stern-brocot* into a loopless algorithm requires some effort.

As a first step towards this goal, let us disentangle *stern-brocot* into a tree of numerators and denominators. Given the specification

$$\text{num} \div \text{den} = \text{stern-brocot}, \tag{8}$$

where  $\div$  is lifted to trees, we reason as follows:

$$\begin{aligned} &\text{num} \div \text{den} \\ = &\quad \{ \text{specification and definition of } \text{stern-brocot} \} \\ &\text{Node } 1 \ (1 / (1 / (\text{num} \div \text{den}) + 1)) \ (\text{num} \div \text{den} + 1) \\ = &\quad \{ \text{arithmetic} \} \\ &\text{Node } (1 \div 1) \ (\text{num} \div (\text{num} + \text{den})) \ ((\text{num} + \text{den}) \div \text{den}) \end{aligned}$$

$$= \{ \text{definition of } \div \text{ lifted to trees } \} \\ (\text{Node } 1 \text{ num } (\text{num} + \text{den})) \div (\text{Node } 1 (\text{num} + \text{den}) \text{ den}).$$

Thus, *num* and *den* defined by

$$\text{num} = \text{Node } 1 \text{ num } (\text{num} + \text{den}) \\ \text{den} = \text{Node } 1 (\text{num} + \text{den}) \text{ den}$$

satisfy the specification. The two definitions are pleasingly symmetric; in fact, we have  $\text{den} = \text{chop num}$ . In other words, we can confine ourselves to linearising *den*; that is we seek to express *chop den* in terms of *den* and possibly *num*. To see what we are aiming for, let us unroll *chop den*:

$$\text{chop den} = \text{Node } 2 \text{ den } (\text{den} + \text{chop den}).$$

This is almost the sum of *num* and *den*:

$$\text{num} + \text{den} - \text{chop den} = \text{Node } 0 (2 * \text{num}) (\text{num} + \text{den} - \text{chop den}) .$$

The difference between *num* + *den* and *chop den* equals  $2 * x$ , where  $x$  is the solution of  $x = \text{Node } 0 \text{ num } x$ . Have we made any progress? Somewhat surprisingly, the answer is yes. By a stroke of good luck, the unique solution  $x$  of the equation above is *num mod den*, as a quick calculation shows:

$$\begin{aligned} & \text{num mod den} \\ = & \{ \text{definition of num and definition of den } \} \\ & (\text{Node } 1 \text{ num } (\text{num} + \text{den})) \text{ mod } (\text{Node } 1 (\text{num} + \text{den}) \text{ den}) \\ = & \{ \text{definition of mod lifted to trees } \} \\ & \text{Node } (1 \text{ mod } 1) (\text{num mod } (\text{num} + \text{den})) ((\text{num} + \text{den}) \text{ mod } \text{den}) \\ = & \{ \text{properties of mod } \} \\ & \text{Node } 0 \text{ num } (\text{num mod den}) . \end{aligned}$$

As an intermediate summary, we have derived

$$\text{fusc} = 1 < \text{fusc}' \\ \text{fusc}' = 1 < \text{fusc} + \text{fusc}' - 2 * (\text{fusc mod fusc}'),$$

where *fusc* = *stream num* is the stream of numerators and *fusc'* = *stream den* = *tail fusc* is the stream of denominators. (The name *fusc* is due to Dijkstra (1976).) Both streams are given by recursive definitions. It is a simple exercise to turn them into iterative definitions. Tupling *fusc* and *fusc'* using  $(*) = \text{zip } (,)$ , we obtain

$$\begin{aligned} & \text{fusc } * \text{fusc}' \\ = & \{ \text{definition of fusc and definition of fusc}' \} \\ & (1 < \text{fusc}') * (1 < \text{fusc} + \text{fusc}' - 2 * (\text{fusc mod fusc}')) \\ = & \{ \text{definition of } * \text{ and definition of zip } \} \\ & (1, 1) < \text{fusc}' * (\text{fusc} + \text{fusc}' - 2 * (\text{fusc mod fusc}')) \end{aligned}$$

$$= \{ \text{idioms, and introduce } \textit{step} (n, d) = (d, n + d - 2 * (n \bmod d)) \} \\ (1, 1) < \textit{map step} (\textit{fusc} * \textit{fusc}')$$

Since *iterate f a* is the unique solution of the recursion equation  $x = a < \textit{map f x}$ , we have  $\textit{fusc} * \textit{fusc}' = \textit{iterate step} (1, 1)$ . The following calculations summarise our findings:

$$\begin{aligned} & \textit{stream stern-brocot} \\ = & \{ \text{see above} \} \\ & \textit{stream} (\textit{num} \div \textit{den}) \\ = & \{ \textit{stream} \text{ is an idiom homomorphism} \} \\ & \textit{stream num} \div \textit{stream den} \\ = & \{ \text{see above} \} \\ & \textit{fusc} \div \textit{fusc}' \\ = & \{ \text{idioms, and introduce } \textit{rat}' (n, d) = n \div d \} \\ & \textit{map rat}' (\textit{fusc} * \textit{fusc}') \\ = & \{ \text{see above} \} \\ & \textit{map rat}' (\textit{iterate step} (1, 1)). \end{aligned}$$

As a final step, we can additionally fuse *rat'* with *iterate*, employing the following formula:

$$1 / ([n \div d] + 1 - \{n \div d\}) = d \div (n + d - 2 * (n \bmod d)).$$

Here,  $[r]$  denotes the integral part of  $r$  and  $\{r\}$  its fractional part, such that  $r = [r] + \{r\}$ . Continuing the calculation, we obtain

$$= \{ \text{fusion, and introduce } \textit{next} r = 1 / ([r] + 1 - \{r\}) \} \\ \textit{iterate next} 1.$$

This intriguing algorithm is attributed to Newman (Aigner & Ziegler 2004); the formula for *step* is apparently due to Stay (Sloane 2009; sequence A002487).

Can we derive a similar algorithm for *stream bird*? The answer is probably no. The next section explains why.

### 6 Linearising the Bird tree and some others

Now that we have warmed up, let's become more ambitious: the goal of this section is to derive a *loopless algorithm* for enumerating the elements of the infinite tree  $ab = \textit{recurse} (a \cdot) (b \cdot) \epsilon$ , where  $a$  and  $b$  are elements of some given monoid. Unfortunately, we will not quite achieve our goal: the step function will run in *amortised* constant time, based on the assumption that the monoid operation  $\cdot$  runs in constant time. Or put differently, it will use a constant number of  $\cdot$  operations amortised over time.

Now, the call *stream ab* yields

$$\epsilon < a < b < a \cdot a < b \cdot a < a \cdot b < b \cdot b < a \cdot a \cdot a < \dots$$

On the face of it, calculating the next element in *stream ab* corresponds to the binary increment:  $b^i \cdot a \cdot w$  becomes  $a^i \cdot b \cdot w$ , and  $b^i$  becomes  $a^{i+1}$ .

In contrast to the binary increment, we can't examine the elements, since we are not working with the free monoid – after all, the elements  $a$  and  $b$  could be functions. Of course, if we don't make any additional assumptions about the underlying structure, then we simply can't calculate the next from the previous element. In order to support incremental computations, we assume that each element has an inverse; that is we are working with a group rather than a monoid:

**class** (*Monoid*  $\alpha$ )  $\Rightarrow$  *Group*  $\alpha$  **where**  
*inverse*  $:: \alpha \rightarrow \alpha$   
 $(\wedge) \quad :: \alpha \rightarrow \text{Integer} \rightarrow \alpha$ .

The class *Group* additionally supports exponentiation, which we assume defaults to a logarithmic implementation. We abbreviate  $a \wedge n$  by  $a^n$ ; in particular, *inverse*  $a = a^{-1}$ .

Now, to get from  $b^i \cdot a \cdot w$  to  $a^i \cdot b \cdot w$ , we simply pre-multiply the former element with  $(a^i \cdot b) \cdot (b^i \cdot a)^{-1}$ . If the current element is  $b^i$ , then we cannot reuse any information, and we compute  $a^{i+1}$  afresh. Still, there is no way to inspect the elements, so we have to maintain some information: the number  $i$  of leading  $bs$  and whether the current element contains only  $bs$ . It turns out that the calculations are slightly more attractive, if we maintain the number of leading  $as$  of the *next* element. Given this information, the next element can be computed as follows:

$$\langle c, i, x \rangle \mid c = a^i \\ \mid \text{otherwise} = a^i \cdot b \cdot a^{-1} \cdot b^{-i} \cdot x.$$

The required pieces of information can be easily defined using infinite trees:

$$\text{rim} = \text{Node True (pure False) rim} \\ \text{carry}' = \text{Node 1 0 (1 + carry')}.$$

Abbreviating  $\text{map } (x \cdot) s$  by  $x \cdot s$ , we have  $ab = \text{Node } \epsilon (a \cdot ab) (b \cdot ab)$ . Only the elements on the right spine of  $ab$  contain only  $bs$ . Consequently, *rim*'s right spine is labelled with *Trues*, and all the other elements are *False*. To motivate the definition of *carry'*, let's unfold *chop ab*:

$$\begin{aligned} \text{chop } ab & \\ = & \quad \{ \text{definition of chop} \} \\ & \quad \text{Node (root (left ab)) (right ab) (chop (left ab))} \\ = & \quad \{ \text{definition of ab} \} \\ & \quad \text{Node } a (b \cdot ab) (\text{chop } (a \cdot ab)) \\ = & \quad \{ \text{chop is an idiom homomorphism} \} \\ & \quad \text{Node } a (b \cdot ab) (a \cdot \text{chop } ab). \end{aligned}$$

The root contains one leading  $a$ , the left subtree none and each element of the right subtree one more than the corresponding element in the entire tree. The definition of *carry'* exactly captures this description.

Lifting the ternary operation  $\langle -, -, - \rangle$  to infinite trees, we claim that  $\langle rim, carry', ab \rangle = chop\ ab$ . The proof makes essential use of the *shift* property

$$\langle c, i + 1, x \rangle = a \cdot \langle c, i, b^{-1} \cdot x \rangle, \tag{9}$$

which expresses the straightforward fact that we can pull an  $a$  to the front if the next element has at least one leading  $a$ . Turning to the proof, we show that  $\langle rim, carry', ab \rangle$  satisfies the same recursion equation as  $chop\ as$ , namely  $x = Node\ a\ (b \cdot ab)\ (a \cdot x)$ :

$$\begin{aligned} & \langle rim, carry', ab \rangle \\ = & \{ \text{definition of } rim, \text{ definition of } carry' \text{ and definition of } ab \} \\ & (Node\ True\ (pure\ False)\ rim, Node\ 1\ 0\ (carry' + 1), Node\ \epsilon\ (a \cdot ab)\ (b \cdot ab)) \\ = & \{ \text{definition of } \langle -, -, - \rangle \text{ lifted to trees} \} \\ & Node\ \langle True, 1, \epsilon \rangle \langle pure\ False, 0, a \cdot ab \rangle \langle rim, carry' + 1, b \cdot ab \rangle \\ = & \{ \text{definition of } \langle -, -, - \rangle \text{ and (9)} \} \\ & Node\ a\ (b \cdot a^{-1} \cdot a \cdot ab)\ (a \cdot \langle rim, carry', b^{-1} \cdot b \cdot ab \rangle) \\ = & \{ \text{inverses} \} \\ & Node\ a\ (b \cdot ab)\ (a \cdot \langle rim, carry', ab \rangle). \end{aligned}$$

So we have reduced the problem of linearising  $ab$  to the problem of linearising  $rim$  and  $carry'$ . Are we any better off? Certainly,  $rim$  isn't difficult to enumerate, but what about  $carry'$ ? By a second stroke of good luck, there is an intriguing cross-connection to the Stern–Brocot tree:  $carry' = \lfloor \text{stern-brocot} \rfloor = \lfloor num / den \rfloor = num\ \mathbf{div}\ den$ . Here is the straightforward proof:

$$\begin{aligned} & num\ \mathbf{div}\ den \\ = & \{ \text{definition of } num \text{ and definition of } den \} \\ & (Node\ 1\ num\ (num + den))\ \mathbf{div}\ (Node\ 1\ (num + den)\ den) \\ = & \{ \text{definition of } \mathbf{div} \text{ lifted to trees} \} \\ & Node\ (1\ \mathbf{div}\ 1)\ (num\ \mathbf{div}\ (num + den))\ ((num + den)\ \mathbf{div}\ den) \\ = & \{ \text{definition of } \mathbf{div} \text{ and } num \geq 1 \leq den \} \\ & Node\ 1\ 0\ ((num\ \mathbf{div}\ den) + 1). \end{aligned}$$

**Exercise 6** Show that  $rim = den == 1$ , where  $==$  is equality lifted to trees, that is  $==$  has type  $(Eq\ \alpha) \Rightarrow Tree\ \alpha \rightarrow Tree\ \alpha \rightarrow Tree\ Bool$ .

In other words, we can reuse the results of the previous section to solve the more general problem of turning  $ab$  into a stream:

$$\begin{aligned} & stream\ ab \\ = & \{ \text{definition of } stream \} \\ & root\ ab\ \prec\ stream\ (chop\ ab) \\ = & \{ \text{see above and Exercise 6} \} \end{aligned}$$

$$\begin{aligned}
& \epsilon < \text{stream } \langle \text{den} == 1, \text{num } \mathbf{div} \text{ den}, ab \rangle \\
& = \quad \{ \text{introduce } \otimes \text{ with } (n, d) \otimes x = \langle d == 1, n \mathbf{div} d, x \rangle \} \\
& \quad \epsilon < \text{stream } ((\text{num} \star \text{den}) \otimes ab) \\
& = \quad \{ \text{stream is an idiom homomorphism } \} \\
& \quad \epsilon < \text{stream } (\text{num} \star \text{den}) \otimes \text{stream } ab \\
& = \quad \{ \text{Section 5 } \} \\
& \quad \epsilon < \text{iterate step } (1, 1) \otimes \text{stream } ab.
\end{aligned}$$

All that is left to do is to express  $\text{stream } ab = \epsilon < \text{iterate step } (1, 1) \otimes \text{stream } ab$  as an iteration. This is easy to achieve using tupling. Let  $q = (1, 1)$ ; then

$$\begin{aligned}
& \text{iterate step } q \star \text{stream } ab \\
& = \quad \{ \text{definition of } \text{iterate} \text{ and property of } \text{stream } ab \} \\
& \quad (q < \text{map step } (\text{iterate step } q)) \star (\epsilon < \text{iterate step } q \otimes \text{stream } ab) \\
& = \quad \{ \text{definition of } \star \text{ and definition of } \text{zip} \} \\
& \quad (q, \epsilon) < \text{map step } (\text{iterate step } q) \star (\text{iterate step } q \otimes \text{stream } ab) \\
& = \quad \{ \text{idioms and introduce } \text{step}'(x, y) = (\text{step } x, x \otimes y) \} \\
& \quad (q, \epsilon) < \text{map step}' (\text{iterate step } q \star \text{stream } ab).
\end{aligned}$$

Recalling that  $\text{iterate } f e$  is the unique solution of the equation  $x = e < \text{map } f x$ , we have established that  $\text{iterate step } q \star \text{stream } ab = \text{iterate step}'(q, \epsilon)$  and furthermore that  $\text{stream } ab = \text{map snd } (\text{iterate step}'(q, \epsilon))$ .

The following definition summarises the derivation – we have additionally inlined the definitions and flattened the nested pair  $(q, \epsilon)$  to a triple:

$$\begin{aligned}
\text{loopless} & \quad :: (\text{Group } \alpha) \Rightarrow \alpha \rightarrow \alpha \rightarrow \text{Stream } \alpha \\
\text{loopless } a \ b & = \text{map } (\lambda(x, y, z) \rightarrow z) (\text{iterate step3 } (1, 1, \epsilon)) \\
& \quad \mathbf{where} \\
& \quad \text{step3 } (n, d, x) = (d, n + d - 2 * m, x') \\
& \quad \quad \mathbf{where } (i, m) = \text{divMod } n \ d \\
& \quad \quad \quad x' \mid d == 1 \quad = a^i \\
& \quad \quad \quad \mid \text{otherwise} = a^i \cdot b \cdot a^{-1} \cdot b^{-i} \cdot x.
\end{aligned}$$

Assuming that the operation ‘ $\cdot$ ’ has a constant running time, the function  $\text{step3}$  takes  $\Theta(\log \log(n + 1))$  steps to produce the  $(n + 1)$ st element from the  $n$ th element: the exponent  $i$  in the definition of  $\text{step3}$  is at most  $\lceil \lg(n + 1) \rceil$ , and fast exponentiation uses at most  $2 \lceil \lg(i + 1) - 1 \rceil$  multiplications. The *amortised running time* of  $\text{step3}$  would be, however, constant, even if exponentiation were implemented naively;  $\text{step3}$  would then perform the same number of steps as the binary increment.

Linearising the Bird tree is now simply a matter of applying  $\text{loopless}$ . First of all, the set of all invertible square matrices forms a group, the so-called general linear group  $GL_n(\mathbb{R})$  – if the coefficients are drawn from  $\mathbb{R}$ . Since  $\begin{bmatrix} 1 & 1 \\ 0 & 1 \end{bmatrix}$  and  $\begin{bmatrix} 1 & 0 \\ 1 & 1 \end{bmatrix}$  have the determinant  $-1$ , they are both invertible in  $GL_2(\mathbb{Z})$ , and we have



$$\begin{aligned}
 & \text{stream bird} \\
 = & \quad \{ \text{Section 3} \} \\
 & \text{stream (map mediant (recurse (■■*) (■■*) I))} \\
 = & \quad \{ \text{stream is an idiom homomorphism} \} \\
 & \text{map mediant (stream (recurse (■■*) (■■*) I))} \\
 = & \quad \{ \text{see above} \} \\
 & \text{map mediant (loopless ■■ ■■)}.
 \end{aligned}$$

Done! Well, not quite: because of the way *stream* is defined, the program above actually enumerates the elements of the bit-reversal permutation tree of *bird*. We should really linearise *recurse (\*■■) (\*■■) I* instead of *recurse (■■\*) (■■\*) I*. Of course, *loopless* can be adapted to work with pre- instead of post-multiplications, but fortunately, there is a more modular approach. Using matrix transposition  $(-)^T$  we can put *recurse (\*■■) (\*■■) I* into the required form:

$$\begin{aligned}
 & \text{stream (map mediant (recurse (*■■) (*■■) I))} \\
 = & \quad \{ I^T = I, \text{■■}^T = \text{■■} \text{ and } \text{■■}^T = \text{■■} \} \\
 & \text{stream (map mediant (recurse (*■■^T) (*■■^T) I^T))} \\
 = & \quad \{ \text{fusion: } (X * Y)^T = Y^T * X^T \} \\
 & \text{stream (map mediant (map transpose (recurse (■■*) (■■*) I)))} \\
 = & \quad \{ \text{functor and see above} \} \\
 & \text{map (mediant } \circ \text{ transpose) (loopless ■■ ■■)}.
 \end{aligned}$$

Can we improve the running time of the final program? If we managed to determine  $X^i$  in constant time, then *loopless* could be turned into a true loopless algorithm. Recall the findings of Section 3:

$$\text{■■}^n = \begin{pmatrix} F_{n-1} & F_n \\ F_n & F_{n+1} \end{pmatrix} \quad \text{■■}^n = \begin{pmatrix} 1 & 0 \\ n & 1 \end{pmatrix} \quad \text{■■}^n = \begin{pmatrix} 1 & n \\ 0 & 1 \end{pmatrix} \quad \text{■■}^n = \begin{pmatrix} F_{n+1} & F_n \\ F_n & F_{n-1} \end{pmatrix}.$$

The Stern–Brocot tree and the Calkin–Wilf tree can indeed be enumerated looplessly, as both involve only ■■ and ■■ – Backhouse and Ferreira derive these special cases in (2008). However, since we can’t compute the Fibonacci numbers in constant time, this doesn’t work for the Bird tree. Indeed, the fastest algorithm for computing  $F_n$  involves calculating ■■<sup>n</sup> using fast exponentiation.

### Acknowledgments

Thanks are due to Roland Backhouse, Jeremy Gibbons, Daniel James and Tom Harper for carefully proofreading a previous version of this paper. I am furthermore grateful to the anonymous referees for many helpful suggestions, in particular, for insisting on a roadmap. Finally, I would like to thank Richard for many years of inspiration and support.

## References

- Aigner, M. & Ziegler, G. M. (2004) *Proofs from THE BOOK*, 3rd edn. Springer-Verlag.
- Backhouse, R. & Ferreira, J. F. (2008) Recounting the rationals: Twice!, In *The 9th International Conference on Mathematics of Program Construction (MPC '08)*, Audebaud, P. & Paulin-Mohring, C. (eds), Lecture Notes in Computer Science, vol. 5133. Springer, pp. 79–91.
- Bird, R. S. (2006) Loopless functional algorithms. In *The 8th International Conference on Mathematics of Program Construction (MPC '06)*, Uustalu, T. (ed.), Lecture Notes in Computer Science, vol. 4014. Springer, pp. 90–114.
- Calkin, N. & Wilf, H. (2000) Recounting the rationals, *Am. Math. Monthly*, 107 (4): 360–363.
- Dijkstra, E. W. (1976) EWD 570: An exercise for Dr. R. M. Burstall. In *Selected Writings on Computing: A Personal Perspective*, Dijkstra, E. W. Springer, pp. 215–216. ISBN 0–387–90652–5.
- Gibbons, J., Lester, D. & Bird, R. (2006) Functional pearl: Enumerating the rationals, *J. Funct. Program.*, 16 (3): 281–291.
- Graham, R. L., Knuth, D. E. & Patashnik, O. (1994) *Concrete Mathematics*, 2nd ed. Addison-Wesley.
- Hinze, R. (2008) Functional pearl: Streams and unique fixed points. In *Proceedings of the 13th ACM SIGPLAN International Conference on Functional Programming (ICFP '08)*, Thiemann, P. (ed.). ACM Press, pp. 189–200.
- McBride, C. & Paterson, R. (2008) Functional pearl: Applicative programming with effects, *J. Funct. Program.*, 18 (1): 1–13.
- Peyton Jones, S. (2003) *Haskell 98 Language and Libraries*. Cambridge University Press.
- Rutten, J. (2003) Fundamental study: Behavioural differential equations: A coinductive calculus of streams, automata, and power series, *Theoret. Comp. Sci.*, 308: 1–53.
- Sloane, N. J. A. (2009) The on-line encyclopedia of integer sequences [online]. Available at: <http://www.research.att.com/~njas/sequences/> (Accessed 17 July 2009).